F02WUF – NAG Fortran Library Routine Document

Note. Before using this routine, please read the Users' Note for your implementation to check the interpretation of bold italicised terms and other implementation-dependent details.

1 Purpose

F02WUF returns all, or part, of the singular value decomposition of a real upper triangular matrix.

2 Specification

```
 SUBROUTINE F02WUF(N, A, LDA, NCOLB, B, LDB, WANTQ, Q, LDQ, SV,
1                  WANTP, WORK, IFAIL)
 INTEGER           N, LDA, NCOLB, LDB, LDQ, IFAIL
 real              A(LDA,*), B(LDB,*), Q(LDQ,*), SV(*), WORK(*)
 LOGICAL           WANTQ, WANTP
```

3 Description

The n by n upper triangular matrix R is factorized as

$$R = QSP^T,$$

where Q and P are n by n orthogonal matrices and S is an n by n diagonal matrix with non-negative diagonal elements, $\sigma_1, \sigma_2, \ldots, \sigma_n$, ordered such that

$$\sigma_1 \geq \sigma_2 \geq \ldots \geq \sigma_n \geq 0.$$

The columns of Q are the left-hand singular vectors of R, the diagonal elements of S are the singular values of R and the columns of P are the right-hand singular vectors of R.

Either or both of Q and P^T may be requested and the matrix C given by

$$C = Q^T B,$$

where B is an n by *ncolb* given matrix, may also be requested.

The routine obtains the singular value decomposition by first reducing R to bidiagonal form by means of Givens plane rotations and then using the QR algorithm to obtain the singular value decomposition of the bidiagonal form.

Good background descriptions to the singular value decomposition are given in Chan [1], Dongarra *et al.* [2], Golub and Van Loan [3], Hammarling [4] and Wilkinson [5].

Note that if K is any orthogonal diagonal matrix so that

$$KK^T = I$$

(that is the diagonal elements of K are $+1$ or -1) then

$$A = (QK)S(PK)^T$$

is also a singular value decomposition of A.

4 References

[1] Chan T F (1982) An improved algorithm for computing the singular value decomposition *ACM Trans. Math. Software* **8** 72–83

[2] Dongarra J J, Moler C B, Bunch J R and Stewart G W (1979) *LINPACK Users' Guide* SIAM, Philadelphia

[3] Golub G H and van Loan C F (1996) *Matrix Computations* Johns Hopkins University Press (3rd Edition), Baltimore

[4] Hammarling S (1985) The singular value decomposition in multivariate statistics *SIGNUM Newsl.* **20 (3)** 2–25

[5] Wilkinson J H (1978) Singular Value Decomposition – Basic Aspects *Numerical Software – Needs and Availability* (ed D A H Jacobs) Academic Press

5 Parameters

1: N — INTEGER *Input*

On entry: n, the order of the matrix R.

When N = 0 then an immediate return is effected.

Constraint: N $\geq$ 0.

2: A(LDA,$*$) — ***real*** array *Input/Output*

Note: the second dimension of the array A must be at least max(1, N).

On entry: the leading n by n upper triangular part of the array A must contain the upper triangular matrix R.

On exit: if WANTP = .TRUE., the n by n part of A will contain the n by n orthogonal matrix P^T, otherwise the n by n upper triangular part of A is used as internal workspace, but the strictly lower triangular part of A is not referenced.

3: LDA — INTEGER *Input*

On entry: the first dimension of the array A as declared in the (sub)program from which F02WUF is called.

Constraint: LDA $\geq$ max(1, N).

4: NCOLB — INTEGER *Input*

On entry: $ncolb$, the number of columns of the matrix B.

When NCOLB = 0 the array B is not referenced.

Constraint: NCOLB $\geq$ 0.

5: B(LDB,$*$) — ***real*** array *Input/Output*

Note: the second dimension of the array B must be at least max(1, NCOLB).

On entry: with NCOLB > 0, the leading n by $ncolb$ part of the array B must contain the matrix to be transformed.

On exit: the leading n by $ncolb$ part of the array B is overwritten by the matrix$Q^T B$.

6: LDB — INTEGER *Input*

On entry: the first dimension of the array B as declared in the (sub)program from which F02WUF is called.

Constraint: when NCOLB > 0 then LDB $\geq$ max(1, N).

7: WANTQ — LOGICAL *Input*

On entry: WANTQ must be .TRUE. if the matrix Q is required. If WANTQ = .FALSE., then the array Q is not referenced.

8: Q(LDQ,*) — ***real*** array *Output*

Note. If WANTQ = .TRUE., then the second dimension of the array Q must be greater than or equal to max(1, N).

On exit: with WANTQ = .TRUE., the leading n by n part of the array Q will contain the orthogonal matrix Q. Otherwise the array Q is not referenced.

9: LDQ — INTEGER *Input*

On entry: the first dimension of the array Q as declared in the (sub)program from which F02WUF is called.

Constraint: if WANTQ = .TRUE., LDQ $\geq$ max(1, N) .

10: SV(*) — ***real*** array *Output*

Note: the dimension of the array SV must be at least max(1, N).

On exit: the array SV will contain the n diagonal elements of the matrix S.

11: WANTP — LOGICAL *Input*

On entry: WANTP must be .TRUE. if the matrix P^T is required, in which case P^T is overwritten on the array A, otherwise WANTP must be .FALSE..

12: WORK(*) — ***real*** array *Output*

Note. The dimension of the array WORK must be greater than or equal to max(1, p) where

$p = 2 \times (\mathrm{N}-1)$	if NCOLB = 0 and WANTQ = .FALSE. and WANTP = .FALSE..
$p = 3 \times (\mathrm{N}-1)$	if (NCOLB = 0 and WANTQ = .FALSE. and WANTP = .TRUE.) or (WANTP = .FALSE. and (NCOLB > 0 and/or WANTQ = .TRUE.))
$p = 5 \times (\mathrm{N}-1)$	otherwise.

On exit: WORK(N) contains the total number of iterations taken by the QR algorithm.

The rest of the array is used as internal workspace.

13: IFAIL — INTEGER *Input/Output*

On entry: IFAIL must be set to 0, −1 or 1. For users not familiar with this parameter (described in Chapter P01) the recommended value is 0.

On exit: IFAIL = 0 unless the routine detects an error (see Section 6).

6 Error Indicators and Warnings

If on entry IFAIL = 0 or −1, explanatory error messages are output on the current error message unit (as defined by X04AAF).

Errors detected by the routine:

IFAIL = −1

On entry, N < 0,
or LDA < N,
or NCOLB < 0,
or LDB < N and NCOLB > 0,
or LDQ < N and WANTQ = .TRUE..

IFAIL > 0

The QR algorithm has failed to converge in 50 × N iterations. In this case SV(1), SV(2), . . . , SV(IFAIL) may not have been found correctly and the remaining singular values may not be the smallest. The matrix R will nevertheless have been factorized as $R = QEP^T$, where E is a bidiagonal matrix with SV(1), SV(2), . . . , SV(n) as the diagonal elements and WORK(1), WORK(2), . . . , WORK($n-1$) as the super-diagonal elements.

This failure is not likely to occur.

7 Accuracy

The computed factors Q, S and P satisfy the relation

$$QSP^T = R + E,$$

where

$$\|E\| \leq c\epsilon\|A\|$$

ϵ is the ***machine precision***, c is a modest function of n and $\|.\|$ denotes the spectral (two) norm. Note that $\|A\| = \mathrm{SV}(1)$.

A similar result holds for the computed matrix $Q^T B$.

The computed matrix Q satisfies the relation

$$Q = T + F,$$

where T is exactly orthogonal and

$$\|F\| \leq d\epsilon$$

where d is a modest function of n. A similar result holds for P.

8 Further Comments

For given values of NCOLB, WANTQ and WANTP, the number of floating-point operations required is approximately proportional to n^3.

Following the use of this routine the rank of R may be estimated by a call to the INTEGER FUNCTION F06KLF. The statement:

```
IRANK = F06KLF(N, SV, 1, TOL)
```

returns the value $(k-1)$ in IRANK, where k is the smallest integer for which $\mathrm{SV}(k) < \mathit{tol} \times \mathrm{SV}(1)$, and l is the tolerance supplied in TOL, so that IRANK is an estimate of the rank of S and thus also of R. If TOL is supplied as negative then the ***machine precision*** is used in place of TOL.

9 Example

To find the singular value decomposition of the 3 by 3 upper triangular matrix

$$A = \begin{pmatrix} -4 & -2 & -3 \\ 0 & -3 & -2 \\ 0 & 0 & -4 \end{pmatrix},$$

together with the vector $Q^T b$ for the vector

$$b = \begin{pmatrix} -1 \\ -1 \\ -1 \end{pmatrix}.$$

9.1 Program Text

Note. The listing of the example program presented below uses bold italicised terms to denote precision-dependent details. Please read the Users' Note for your implementation to check the interpretation of these terms. As explained in the Essential Introduction to this manual, the results produced may not be identical for all implementations.

```
*     F02WUF Example Program Text
*     Mark 14 Release.  NAG Copyright 1989.
*     .. Parameters ..
      INTEGER          NIN, NOUT
      PARAMETER        (NIN=5,NOUT=6)
      INTEGER          NMAX, NCOLB, LDA, LDB, LDQ, LWORK
```

```
      PARAMETER        (NMAX=10,NCOLB=1,LDA=NMAX,LDB=NMAX,LDQ=NMAX,
     +                 LWORK=5*(NMAX-1))
*     .. Local Scalars ..
      INTEGER          I, IFAIL, J, N
      LOGICAL          WANTP, WANTQ
*     .. Local Arrays ..
      real             A(LDA,NMAX), B(LDB), Q(LDQ,NMAX), SV(NMAX),
     +                 WORK(LWORK)
*     .. External Subroutines ..
      EXTERNAL         F02WUF
*     .. Executable Statements ..
      WRITE (NOUT,*) 'F02WUF Example Program Results'
*     Skip heading in data file
      READ (NIN,*)
      READ (NIN,*) N
      WRITE (NOUT,*)
      IF (N.GT.NMAX) THEN
         WRITE (NOUT,*) 'N is out of range.'
         WRITE (NOUT,99999) 'N = ', N
      ELSE
         READ (NIN,*) ((A(I,J),J=I,N),I=1,N)
         READ (NIN,*) (B(I),I=1,N)
         WANTQ = .TRUE.
         WANTP = .TRUE.
         IFAIL = 0
*
*        Find the SVD of A
         CALL F02WUF(N,A,LDA,NCOLB,B,LDB,WANTQ,Q,LDQ,SV,WANTP,WORK,
     +               IFAIL)
*
         WRITE (NOUT,*) 'Singular value decomposition of A'
         WRITE (NOUT,*)
         WRITE (NOUT,*) 'Singular values'
         WRITE (NOUT,99998) (SV(I),I=1,N)
         WRITE (NOUT,*)
         WRITE (NOUT,*) 'Left-hand singular vectors, by column'
         DO 20 I = 1, N
            WRITE (NOUT,99998) (Q(I,J),J=1,N)
   20    CONTINUE
         WRITE (NOUT,*)
         WRITE (NOUT,*) 'Right-hand singular vectors, by column'
         DO 40 I = 1, N
            WRITE (NOUT,99998) (A(J,I),J=1,N)
   40    CONTINUE
         WRITE (NOUT,*)
         WRITE (NOUT,*) 'Vector Q''*B'
         WRITE (NOUT,99998) (B(I),I=1,N)
      END IF
      STOP
*
99999 FORMAT (1X,A,I5)
99998 FORMAT (3(1X,F8.4))
      END
```

9.2 Program Data

```
F02WUF Example Program Data
  3                :Value of N
 -4.0  -2.0  -3.0
       -3.0  -2.0
             -4.0   :End of matrix A
 -1.0  -1.0  -1.0   :End of vector B
```

9.3 Program Results

```
F02WUF Example Program Results

Singular value decomposition of A

Singular values
  6.5616   3.0000   2.4384

Left-hand singular vectors, by column
 -0.7699   0.5883  -0.2471
 -0.4324  -0.1961   0.8801
 -0.4694  -0.7845  -0.4054

Right-hand singular vectors, by column
  0.4694  -0.7845   0.4054
  0.4324  -0.1961  -0.8801
  0.7699   0.5883   0.2471

Vector Q'*B
  1.6716   0.3922  -0.2276
```